

SEARCH FOR HOMOGENEOUS POLYNOMIAL INVARIANTS AND A CUBIC-HOMOGENEOUS MAPPING WITHOUT QUADRATIC INVARIANTS

BY RICARDO DOS SANTOS FREIRE JR., GIANLUCA GORNI AND GAETANO
ZAMPIERI

Abstract. The class of the cubic-homogenous mappings with nonzero constant Jacobian determinant is interesting because if it were proved that all mappings in this class are invertible, then the general Jacobian Conjecture would follow. A secondary conjecture was that all mappings in this class had linear invariants. De Bondt has recently put that to rest, with an example that has no linear invariants. Still, de Bondt's mapping has quadratic invariants. In this paper we exhibit an example in dimension 11 that has a cubic invariant but no quadratic (or linear) ones. We also provide some simple *Mathematica* code that has supported us in our search, and that may still be useful in further investigations.

1. Introduction. The celebrated Jacobian Conjecture asserts that all polynomial mappings of \mathbb{C}^n with everywhere invertible Jacobian matrix are themselves invertible. Starting most notably with Bass, Connell and Wright [1], people have found some special classes of polynomial mappings with the following property: if we prove that all mappings in the class are invertible, then the full Jacobian Conjecture is true. One of these classes, introduced by Yagzhev [5] and, independently, by Bass, Connell and Wright, is made of the so-called “cubic-homogeneous,” or *Yagzhev* maps, which are the maps $f: \mathbb{C}^n \rightarrow \mathbb{C}^n$ of the form

$$(1) \quad f(x) = x - g(x) \quad \text{such that } \det f'(x) = 1 \text{ for all } x \in \mathbb{C}^n,$$

1991 *Mathematics Subject Classification.* 14R15.

Key words and phrases. Cubic-homogeneous maps, linear dependence problem, homogeneous polynomial invariants.

This research was done while Ricardo Freire was at the Dipartimento di Informatica of the University of Verona, Italy, supported by Assegno di Ricerca AdR942/08.

where $g: \mathbb{C}^n \rightarrow \mathbb{C}^n$ is a polynomial mapping *homogeneous of degree 3*. The important subclass of the cubic-linear maps was introduced by Drużkowski [3]. For more results and conjectures on these classes maps we refer to van den Essen's book [4].

What we are concerned here are the *invariants*. We say that the (nonconstant) scalar function $k: \mathbb{C}^n \rightarrow \mathbb{C}$ is an invariant for a mapping $f: \mathbb{C}^n \rightarrow \mathbb{C}^n$ if $k \circ f = k$. Until recently, all known examples of Yagzhev mappings had linear invariants, and so it was conjectured that all mappings in the class have linear invariants too ("linear dependence conjecture"). However, De Bondt in his 2006 paper [2] exhibited a Yagzhev mapping that has no linear invariants.

Zampieri [6] noticed that Bondt's example still has quadratic and cubic invariants, and that these invariants are very helpful in computing the inverse mapping. So it seems relevant to the Jacobian Conjecture to ask whether there are Yagzhev mappings that have no quadratic, or cubic polynomial invariants.

This paper gives a first answer in Section 2 by providing an example of a Yagzhev mapping that has a homogeneous polynomial invariant of degree 3 but has neither linear nor quadratic invariants.

In Section 3 we give the listing of a *Mathematica* program that we used extensively in our investigation; it may still be helpful in the future.

2. A cubic-homogeneous mapping without quadratic invariants.

Let us consider the following nilpotent matrix depending on two parameters:

$$(2) \quad M(s, t) := \begin{pmatrix} 0 & 0 & s^2 \\ 0 & 0 & t^2 \\ -t^2 & s^2 & 0 \end{pmatrix}.$$

Define $g: \mathbb{C}^{11} \rightarrow \mathbb{C}^{11}$ in column form as

$$(3) \quad g(x_1, x_2, \dots, x_{11}) = \begin{pmatrix} M(x_{10}, x_{11}) \begin{pmatrix} x_1 \\ x_2 \\ x_3 \end{pmatrix} \\ M(x_{10}, x_{11}) \begin{pmatrix} x_4 \\ x_5 \\ x_6 \end{pmatrix} \\ M(x_{10}, x_{11}) \begin{pmatrix} x_7 \\ x_8 \\ x_9 \end{pmatrix} \\ \det \begin{pmatrix} x_1 & x_2 & x_3 \\ x_4 & x_5 & x_6 \\ x_7 & x_8 & x_9 \end{pmatrix} \\ x_{10}^3 \end{pmatrix}.$$

For example, the first three components of g are the row-by-column product of the matrix $M(x_{10}, x_{11})$ with the column of the variables x_1, x_2, x_3 . It is clear that g is a polynomial mapping whose components are homogeneous of degree 3 in the 11 variables x_1, \dots, x_{11} .

The example that we announced is the $f: \mathbb{C}^{11} \rightarrow \mathbb{C}^{11}$ defined as

$$(4) \quad f(x) := x - g(x).$$

This mapping has the following k as a cubic invariant:

$$(5) \quad k(x) := \det \begin{pmatrix} x_1 & x_2 & x_3 \\ x_4 & x_5 & x_6 \\ x_7 & x_8 & x_9 \end{pmatrix}$$

as it is not too hard to check even by hand. Using this fact and the nilpotence of M , one can compute the inverse of f , which is also a polynomial mapping. Knowing that f is invertible, we can deduce that its Jacobian determinant must be 1. Of course, a computer can perform these calculations too. It was actually the computer that informed us first that there are no linear and no quadratic homogeneous polynomials which are invariants for f . However, we also managed to find a proof that can be made by pen and paper, and we are going to outline it below.

THEOREM 1. *The function f defined in (4) is a polynomial automorphism of \mathbb{C}^{11} which admits the cubic homogeneous polynomial invariant k defined in (5). Moreover, there are no homogeneous polynomial invariants of degree 1 or 2 for f .*

PROOF. We are only left to prove the absence of linear and quadratic invariants. Actually, since the square of a linear invariant is a quadratic invariant, it is enough to deal with the degree 2. Suppose then that A is an 11×11 symmetric complex matrix, and suppose that the quadratic form $K(x) := x^T A x$ (where the exponent T means transposition) is an invariant for f . By separating the homogeneous parts of the identity $K \circ f = K$, we deduce that

$$(6) \quad x^T A g(x) = 0, \quad g(x)^T A g(x) = 0, \quad \text{for all } x \in \mathbb{C}^{11}.$$

Next, consider the polynomial curve $c(t) = (t^{2^1}, t^{2^2}, \dots, t^{2^{11}})$. The first equation in (6) implies that

$$(7) \quad c(t)^T A g(c(t)) = 0, \quad \text{for all } t \in \mathbb{R}.$$

By means of this condition only, we can prove that A vanishes performing a boring calculation. \square

3. A *Mathematica* program. In our search for examples of polynomial mappings without low-degree invariants, we used the computer algebra system *Mathematica*, by Wolfram Research Inc., version 5.0 or later. The following routine does the heavy work:

```

HomogeneousInvariantSearch [polynomials_List ,
  variables_List , degreeOfInvariant_Integer] :=
Module[{n, degreeOfPolynomials, t, a,
  arrayOfCoefficientsOfInvariant ,
  coefficientsOfInvariant ,
  invariant , invariantComposedWithTheMapping ,
  mustVanish ,
  derivativesAtZero , sol , result },
n = Length[variables];
degreeOfPolynomials =
  Max@Exponent[Collect[polynomials /.
    Thread[variables -> t*variables],
    t], t];
arrayOfCoefficientsOfInvariant =
  Array[a, Evaluate@Table[n,
    {degreeOfInvariant}]] /.
  a[i_ -] /; Not@OrderedQ[{i}] :>
  a @@ Sort[{i}];
coefficientsOfInvariant =
  Union@Flatten@
    arrayOfCoefficientsOfInvariant;
invariant =
  Dot[arrayOfCoefficientsOfInvariant ,
    Sequence @@ Table[variables ,
      {degreeOfInvariant}]];
invariantComposedWithTheMapping =
  Dot[arrayOfCoefficientsOfInvariant ,
    Sequence @@ Table[polynomials ,
      {degreeOfInvariant}]];
mustVanish =
  Expand[invariantComposedWithTheMapping -
    invariant];
derivativesAtZero[0] = {mustVanish};
derivativesAtZero[k_] :=
  derivativesAtZero[k] =
    Union@Flatten[Evaluate@Outer[D,
      derivativesAtZero[k - 1],
      variables] /.
      sol[k - 1]];

```

```

sol[k_] := sol[k] =
  Solve[0 == derivativesAtZero[k] /.
    Thread[variables -> 0],
    coefficientsOfInvariant];
result =
  First@Flatten@Fold[ReplaceAll,
    invariant, Table[sol[n],
      {n, degreeOfPolynomials*
        degreeOfInvariant}]];
If[Simplify@result == 0, 0,
  ColumnForm@Complement[
    Factor@Together@Flatten@
      CoefficientList[result,
        coefficientsOfInvariant],
    {0}]]];

```

It takes as arguments the list of the components of the polynomial map, then the list of the variables, and finally the degree of the desired homogeneous invariants to be found. In the sequel we are going to apply the routine to the 9-dimensional automorphism, called `map9` below, which was considered in [6] as a slight modification of de Bondt's 10-dimensional mapping, and to our 11-dimensional mapping in Section 2, called `map11` below.

```

var = {x1, x2, x3, x4, x5, x6, x7, x8, x9};
map9 =

```

```

  var +
  {x1 x7 x9 + x2 x9^2,
   -x1 x7^2 - x2 x7 x9,
   x3 x7 x9 + x4 x9^2,
   -x3 x7^2 - x4 x7 x9,
   x5 x7 x9 + x6 x9^2,
   -x5 x7^2 - x6 x7 x9,
   (x1 x4 - x2 x3) x9,
   (x3 x6 - x4 x5) x9,
   (x1 x4 - x2 x3) x8 -
   (x3 x6 - x4 x5) x7};

```

```

HomogeneousInvariantSearch[map9, var, 1]

```

```

HomogeneousInvariantSearch[map9, var, 2]

```

The application gives the output

0

for the first call of the routine, and

$$\begin{aligned} &2 (x_2 x_3 - x_1 x_4) \\ &2 (x_2 x_5 - x_1 x_6) \\ &2 (x_4 x_5 - x_3 x_6) \end{aligned}$$

for the second one. Now, let us go to the mapping `map11`:

```
var = {x1, x2, x3, x4, x5, x6,
      x7, x8, x9, x10, x11};
map11 = var -
  {x10^2*x3,
   x11^2*x3,
   x10^2*x2 - x11^2*x1,
   x10^2*x6,
   x11^2*x6,
   x10^2*x5 - x11^2*x4,
   x10^2*x9,
   x11^2*x9,
   x10^2*x8 - x11^2*x7,
   Det[{{x1, x2, x3},
        {x4, x5, x6},
        {x7, x8, x9}}],
   x10^3};
HomogeneousInvariantSearch [map11, var, 1]
HomogeneousInvariantSearch [map11, var, 2]
HomogeneousInvariantSearch [map11, var, 3]
```

The application gives the output

0

for the first and second calls of the routine, and

$$\begin{aligned} &6 (x_3 x_5 x_7 - x_2 x_6 x_7 - \\ &\quad x_3 x_4 x_8 + x_1 x_6 x_8 + \\ &\quad x_2 x_4 x_9 - x_1 x_5 x_9) \end{aligned}$$

for the third one.

References

1. Bass H., Connell E., Wright D., *The Jacobian Conjecture: reduction of degree and formal expansion of the inverse*, Bull. Amer. Math. Soc., **7** (1982), 287–330.
2. de Bondt M., *Quasi-translations and counterexamples to the homogeneous dependence problem*, Proc. Amer. Math. Soc., **134** (2006), 2849–2856.

3. Drużkowski L., *An effective approach to Keller's Jacobian conjecture*, Math. Ann., **264** (1983), 303–313.
4. van den Essen A., *Polynomial Automorphisms*, Progress in Mathematics, **190** (2000), Birkhäuser.
5. Yagzhev A., *Keller's problem*, Siberian Math. J., **21** (1980), 747–754.
6. Zampieri G., *Homogeneous polynomial invariants for cubic-homogeneous functions*, Univ. Iagel. Acta Math., **46** (2008), 99–103.

Received April 21, 2008

Departamento de Matemática Aplicada
Rua do Matão 1010
Cep 05508-090 São Paulo – SP, Brazil
e-mail: rfreire@cecm.usp.br

Dipartimento di Matematica e Informatica
Università di Udine
I-33100 Udine UD, Italy
e-mail: gianluca.gorni@dimi.uniud.it

Dipartimento di Informatica
Università di Verona
Strada Le Grazie, 15
I-37134 Verona, Italy
e-mail: gaetano.zampieri@univr.it